

Keine Angst vor Microsoft Excel! Formeln verstehen, selber aufbauen und erfolgreich einsetzen

von
Hildegard Hügemann, Dieter Schiecke

1. Auflage

Keine Angst vor Microsoft Excel! Formeln verstehen, selber aufbauen und erfolgreich einsetzen – Hügemann / Schiecke
schnell und portofrei erhältlich bei beck-shop.de DIE FACHBUCHHANDLUNG

Thematische Gliederung:

Tabellenkalkulation

Microsoft 2012

Verlag C.H. Beck im Internet:
www.beck.de

ISBN 978 3 86645 559 7

4

Keine Angst vor Logik: Daten auswerten mit WENN & Co.

Mit WENN Berechnungen in eine bestimmte Richtung leiten	114
Fehlerquellen ausschließen mit WENN sowie WENNFEHLER	119
Statusinformationen liefern mit WENN	124
Mehr als zwei mögliche Ergebnisse durch Verschachteln von WENN	125
Noch mehr Bedingungen definieren mit UND, ODER und NICHT	128

Im vorangegangenen Kapitel konnten Sie erfahren, wie Sie mit SUMMEWENN(S), MITTELWERTWENN(S) und ZÄHLENWENN(S) eine Summe, einen Mittelwert oder eine Anzahl nur unter der Bedingung berechnen lassen, dass ein oder mehrere Kriterien erfüllt sind. Doch nicht nur bei Summen oder Durchschnittswerten ist es erforderlich, dass bestimmte Bedingungen zutreffen, bevor gerechnet wird.

Lesen Sie in diesem Kapitel, welche Funktionen noch flexibler sind und wie Sie beliebige Berechnungen steuern oder Ergebnisse nach bestimmten Regeln überwachen. All das erledigen Sie mit solchen Logik-Funktionen wie WENN, UND, ODER sowie NICHT.

Mit WENN Berechnungen in eine bestimmte Richtung leiten

Ihr wichtigster Helfer in Sachen Excel-Logik ist die Funktion WENN. Sie wird angewendet, um das Ausführen einer Aktion von einer Bedingung abhängig zu machen. Sie wird daher auch WENN-DANN-SONST-Funktion genannt.

Wie diese Funktion arbeitet, können Sie an einem Beispiel aus dem täglichen Leben studieren, das in Abbildung 4.1 skizziert ist.

Abbildung 4.1 Die WENN-Funktion am Beispiel einer Situation aus dem Alltag erklärt



Wenn das Wetter schön ist (*Prüfung* der Bedingung: »Ist das Wetter schön?«),

- dann gehen wir wandern (*Dann_Wert*: »Wir gehen wandern«),
- sonst bleiben wir zu Hause (*Sonst_Wert*: »Wir bleiben zu Hause«).

Syntax und Einsatzgebiete von WENN

Wie jede andere Funktion in Excel hat auch die WENN-Funktion einen bestimmten Aufbau. Diese Syntax ist wie folgt: =WENN(*Prüfung*; *Dann_Wert*; *Sonst_Wert*).

- Mit WENN prüfen Sie also, ob bestimmte Bedingungen erfüllt sind (im oben genannten Beispiel »Ist das Wetter schön?«).

- Das Ergebnis einer *Prüfung* ergibt WAHR oder FALSCH.
- Wird eine Bedingung erfüllt (das Ergebnis der Prüfung ist WAHR), folgt der *Dann_Wert* (»Wir gehen wandern«).
- Wird die Bedingung nicht erfüllt (die Prüfung ergibt FALSCH), folgt der *Sonst_Wert* (»Wir bleiben zu Hause«).

Natürlich gibt es viele weitere Fälle, in denen Sie die WENN-Funktion in Excel verwenden – egal ob zu beruflichen oder privaten Zwecken. Hier vier Beispiele:

- Es liegen in zwei Spalten geplante und tatsächliche Kosten vor. In einer dritten Spalte möchten Sie mit einer Meldung gewarnt werden, wenn die Ist- über den Planwerten liegen.
- Beim Einfahren einer neuen Maschine wird alle fünf Minuten die Temperatur geprüft. Die Temperaturwerte werden täglich in Excel importiert. Der Qualitätsprüfer möchte in einer gesonderten Spalte rechts neben den Werten stets dann einen Warnhinweis sehen, wenn Temperaturwerte eine zuvor definierte Obergrenze um mehr als 20% überschreiten.
- Bei Reisekostenabrechnungen oder beim Erstellen von Rechnungen soll automatisch der richtige Mehrwertsteuersatz von 7% oder 19% gewählt und anschließend die Beträge für Mehrwertsteuer und Brutto korrekt berechnet werden.
- Im Verein zahlen die Mitglieder ihren Beitrag je nach Altersstufe. Der Kassenswart möchte automatisch einen Hinweis erhalten, wenn ein Mitglied in eine neue Altersstufe kommt und somit ein anderer Beitrag fällig wird.

Table 4.1 Operatoren zur Prüfung der Bedingung

Operator	Bedeutung	Beispiel mit Prüfung; Dann_Wert; Sonst_Wert
=	gleich	Ausgaben = Einnahmen; Ausgeglichen; Prüfen
<	kleiner	Ausgaben < Einnahmen; Geld vorhanden; Prüfen
>	größer	Ausgaben > Einnahmen; Budget überschritten; OK
<=	kleiner gleich	Ausgaben <= Einnahmen; OK; Budget überschritten
>=	größer gleich	Ausgaben >= Einnahmen; Budget ausgeschöpft/überschritten; OK
<>	ungleich	Ausgaben <> Einnahmen; Prüfen; Ausgeglichen

Mit WENN den richtigen Mehrwertsteuersatz finden

Lernen Sie zu Beginn an einem kleinen Beispiel kennen, wie Sie mit der WENN-Funktion den passenden Steuersatz von 7% oder 19% ermitteln und auf der Basis dann den Bruttobetrag korrekt berechnen.



Abbildung 4.2 Vorschau auf die fertige Lösung: den passenden Mehrwertsteuersatz zur Bruttoberechnung finden

	A	B	C	D	E	F	G
8							
9					Steuersatz 1	7%	
10					Steuersatz 2	19%	
11							
12	Leistung	Art	Stückpreis	Anzahl	Netto	MwSt.	Brutto
13	Artikel zu Excel	Publikation	150	3	450 €	7%	483,50 €
14	Vorlage erstellen	Dienstleistung	500	1	500 €	19%	595,00 €
15	Datenbank-Programmierung	Dienstleistung	55	9	495 €	19%	589,05 €
16	Training Office	Training	600	2	1.200 €	19%	1.428,00 €
17	Artikel zu Access	Publikation	170	4	680 €	7%	727,60 €
18	Wartung Datenbank	Dienstleistung	55	2	110 €	19%	130,90 €



Um die Aufgabe zur Berechnung des Bruttobetrags mit dem richtigen Steuersatz nachzuvollziehen, öffnen Sie die Datei *Kap_04_Logik_UEB.xlsx*. Wechseln Sie dort zum Arbeitsblatt *WENN_DANN_SONST*. Ermitteln Sie in der Spalte F mit der WENN-Funktion den richtigen Steuersatz. Führen Sie die Berechnung des Bruttobetrags in Spalte G durch.

Die Liste der Leistungen vorbereiten



Wandeln Sie zunächst die Liste mit den Leistungen mit der Tastenkombination **[Strg]+[T]** in eine »intelligente« Tabelle um. Geben Sie der Tabelle anschließend den Namen *Leistungen*. Das erledigen Sie über die Registerkarte *Tabellentools/Entwurf* ganz links im Feld unter *Tabellenname*.

Abbildung 4.3 Die Leistungstabelle mit Namen versehen



Korrekte Wahl der Leistungsart per Mausclick

Führen Sie die folgenden Schritte aus, um die Leistungsart flexibel und fehlerfrei eingeben zu können:

1. Bewegen Sie im Blatt *WENN_DANN_SONST* die Maus über die Spalte *Art* – der Mauszeiger wird zum Spaltenmarkierer (kleiner schwarzer Pfeil nach unten). Markieren Sie die gesamte Tabellenspalte durch einen Mausclick.
2. Klicken Sie auf der Registerkarte *Daten* auf den oberen Bereich der Schaltfläche *Datenüberprüfung*.
3. In dem in *Abbildung 4.4* gezeigten Dialogfeld *Datenüberprüfung* wählen Sie auf der Registerkarte *Einstellungen* unter *Zulassen* den Eintrag *Liste*.



- Klicken Sie anschließend darunter in das Feld *Quelle*. Tragen Sie dort jeweils durch Semikolon getrennt die Werte *Publikation*, *Dienstleistung* und *Training* ein. Schließen Sie die Aktion mit *OK* ab.

Abbildung 4.4 Liste mit festen Werten für die Leistungsarten vorgeben



Die Mehrwertsteuersätze an einer zentralen Stelle festlegen

Verschiedene Leistungen werden zu unterschiedlichen Steuersätzen erbracht:

- Publikationen mit einem Steuersatz von 7%,
- alle anderen mit einem Steuersatz von 19%.

Hinterlegen Sie die beiden Steuersätze oberhalb der Leistungstabelle an einem zentralen Ort. Dies hat den Vorteil, dass Sie künftig bei eventuellen Steuersatzänderungen schnell die Werte ändern können.

Abbildung 4.5 Steuersätze an einem zentralen Ort hinterlegen

	E	F
9	Steuersatz 1	7%
10	Steuersatz 2	19%

Die Formeln, die sich später auf die hier hinterlegten Steuersätze beziehen, werden lesbarer, wenn Sie die Zelle F9 mit dem Namen *Steuersatz1* und die Zelle F10 mit dem Namen *Steuersatz2* versehen. Markieren Sie dazu die jeweilige Zelle, wählen Sie auf der Registerkarte *Formeln* die Schaltfläche *Namen definieren* und vergeben Sie den jeweiligen Namen.




Den passenden Mehrwertsteuersatz zuweisen

Nach diesen vorbereitenden Arbeiten verwenden Sie nun die WENN-Funktion, um in der Spalte *MwSt.* den passenden Mehrwertsteuersatz zur jeweiligen Leistungsart zu ermitteln. Prüfen Sie dazu, ob in der Spalte *Art* der Eintrag *Publikation* ausgewählt wurde. In diesem Fall wählen Sie *Steuersatz 1*, sonst *Steuersatz 2*.

Gehen Sie wie folgt vor:


1. Klicken Sie in der Tabelle *Leistungen* auf Zelle F13.
2. Wählen Sie nach Eingabe des Gleichheitszeichens = und nach Eintippen des Buchstabens W die Funktion *WENN* aus.

Abbildung 4.6 Syntax der WENN-Funktion



3. Zur Eingabe des Arguments *Prüfung* klicken Sie in Zelle B13, geben den Vergleichsoperator = ein, dann ein Anführungszeichen ", den festen Text *Publikation* und noch einmal ein Anführungszeichen ". Die Formel lautet nun: =WENN([@Art]="Publikation"
4. Schließen Sie das Argument *Prüfung* durch Eingabe eines Semikolons ab.

Abbildung 4.7 Dann_Wert für WENN-Funktion bestimmen



5. Klicken Sie nun die Zelle F9 an. Falls Sie keinen Namen für diese Zelle definiert haben, drücken Sie die [F4]-Taste, um die Zelladresse zu einem absoluten Bezug zu machen. Haben Sie – wie im Tipp oben empfohlen – die Zelle benannt, so können Sie diesen Schritt sparen.
6. Schließen Sie das Argument *Dann_Wert* durch Eingabe eines Semikolons ab.

Abbildung 4.8 Sonst_Wert für WENN-Funktion bestimmen



7. Klicken Sie nun Zelle F10 an, drücken Sie wieder die [F4]-Taste, falls Sie keinen Namen für diese Zelle definiert haben.
8. Schließen Sie die Formel mit einer schließenden runden Klammer ab.

Die Formel lautet nun:

=WENN([@Art]="Publikation";Steuersatz1;Steuersatz2)

Abbildung 4.9 Die WENN-Funktion wählt nur für die Leistungsart *Publikation* den *Steuersatz1* aus, sonst *Steuersatz2*


```
=WENN([@Art]="Publikation";Steuersatz1;Steuersatz2)
```

Die Spalte F beinhaltet nun den zur Leistung passenden Mehrwertsteuersatz wie in Abbildung 4.2 zu sehen.

Den Bruttobetrag mit dem passenden Mehrwertsteuersatz berechnen

Nachdem Sie den passenden Mehrwertsteuersatz ermittelt haben, fällt das Berechnen des Bruttobetrag nicht mehr schwer.

Führen Sie dazu folgende Schritte aus:

1. Klicken Sie die Zelle G13 an.
2. Geben Sie ein Gleichheitszeichen = ein.
3. Markieren Sie die Zelle E13, geben Sie das Pluszeichen + ein.
4. Markieren Sie nun noch einmal die Zelle E13, geben Sie das Multiplikationszeichen * ein und markieren Sie die Zelle F13.
5. Schließen Sie die Formel mit  ab.

Die Formel lautet `=[@Netto]+[@Netto]*[@MwSt.]`. Das Ergebnis für Zelle G13 ist 481,50 €. Dank der Formatierung des Eingabebereichs als »intelligente« Tabelle setzt sich die Berechnung automatisch über die gesamte *Brutto*-Spalte fort. Die Ergebnisse der *Brutto*-Spalte sehen Sie in Abbildung 4.2.

Fehlerquellen ausschließen mit WENN sowie WENNFEHLER

Beim Aufbau einer Tabelle sind zu Beginn einige Zellen noch leer und werden erst im Laufe der Zeit gefüllt. Trotz fehlender Werte werden schon Formeln aufgestellt, doch dies führt nicht selten zu unschönen Fehlermeldungen wie `#DIV/0!` oder `#NV`.

Daher sollen Berechnungen nur dann ausgeführt werden, wenn auch die erforderlichen Werte vorhanden sind. Somit muss der eigentlichen Berechnung eine Abfrage vorangestellt werden, die prüft, ob die zu berechnenden Zellen leer sind, und dann entsprechend handelt.

Außerdem sollen Fälle wie eine Division durch null ausgeschlossen werden.

Sinnvolle Kostenberechnungen durch Prüfung auf Vollständigkeit



In einer Projektplanung werden unter anderem Soll- und Ist-Kosten hinterlegt. Aus diesen beiden Angaben sollen Kostendifferenzen und prozentuale Berechnungen erfolgen, um die Projekte in ihrer Kostenentwicklung zu überwachen. Fehlt an dieser Stelle eine der Angaben, so sind die Berechnungen nicht aussagekräftig oder führen zu Fehlaussagen. Um dies zu verhindern, können mithilfe der WENN-Funktion Zellen auf leere Einträge geprüft und gegebenenfalls die Berechnung verhindert werden.

Abbildung 4.10 VORHER: bei fehlenden Angaben bei den Ist-Kosten werden in den Spalten F und H irritierende Differenzen angezeigt

	A	B	C	D	E	F	G	H
10	Projekt	MA	Plan-Beginn	Soll-Kosten	Ist-Kosten	Diff. abs.	Kosten %	Sum. Diff.
11	Office-Umstellung	12	01.03.2012	254.000 €	238.954 €	-15.046 €	94%	-15.046 €
12	Virtualisierung	4	05.03.2012	66.000 €	0 €	-66.000 €	0%	-66.000 €
13	TK-Anlage	6	06.03.2012	114.500 €	120.664 €	6.164 €	105%	-74.882 €
14	Project-Umstellung	2	08.03.2012	32.400 €	32.400 €	0 €	100%	-74.882 €
15	Windows-Umstellung	5	10.03.2012	93.800 €	108.400 €	14.600 €	116%	-60.282 €
16	Intranet-Update	3	14.03.2012	67.600 €		-67.600 €	0%	-127.682 €
17	SAP-Umstellung	11	15.03.2012	342.100 €	362.743 €	20.643 €	106%	-107.099 €
18	Datensicherheit	7	25.03.2012	159.300 €	160.230 €	930 €	101%	-106.109 €
19	Datenbank-Update	2	28.03.2012	33.490 €		-33.490 €	0%	-139.599 €

Abbildung 4.11 NACHHER: mithilfe der WENN-Funktion werden fehlende Daten beim Berechnen der Differenzen abgefangen und irritierende Werte bei der Projektüberwachung vermieden

	A	B	C	D	E	F	G	H
10	Projekt	MA	Plan-Beginn	Soll-Kosten	Ist-Kosten	Diff. abs.	Kosten %	Sum. Diff.
11	Office-Umstellung	12	01.03.2012	254.000 €	238.954 €	-15.046 €	94%	-15.046 €
12	Virtualisierung	4	05.03.2012	66.000 €	0 €	-66.000 €	0%	-66.000 €
13	TK-Anlage	6	06.03.2012	114.500 €	120.664 €	6.164 €	105%	-74.882 €
14	Project-Umstellung	2	08.03.2012	32.400 €	32.400 €	0 €	100%	-74.882 €
15	Windows-Umstellung	5	10.03.2012	93.800 €	108.400 €	14.600 €	116%	-60.282 €
16	Intranet-Update	3	14.03.2012	67.600 €				
17	SAP-Umstellung	11	15.03.2012	342.100 €	362.743 €	20.643 €	106%	-89.699 €
18	Datensicherheit	7	25.03.2012	159.300 €	160.230 €	930 €	101%	-88.709 €
19	Datenbank-Update	2	28.03.2012	33.490 €				



Um die Prüfung auf Vollständigkeit der Daten nachvollziehen zu können, öffnen Sie die Datei *Kap_04_Logik_UEB.xlsx*. Wechseln Sie dort zum Arbeitsblatt *Fehlerprüfung mit WENN*. Ermitteln Sie in Spalte F mit der WENN-Funktion die Differenz der Ist- und Soll-Kosten und in Spalte G die aktuellen Ist-Kosten als prozentualen Wert im Vergleich zu den Soll-Kosten.

Ist die Differenz zwischen Soll- und Ist-Kosten negativ oder positiv?

In Spalte F *Diff. abs.* soll die Differenz zwischen den aktuellen Ist- und Soll-Kosten darüber Aufschluss geben, ob die Ist-Kosten im geplanten Rahmen liegen. Das Problem hierbei sind die fehlenden Ist-Kosten bei manchen Projekten. Fehlt diese Angabe, so wird die *Diff. abs.* negativ und führt zu einer falschen Aussage. Mithilfe der WENN-Funktion werden diese Fälle abgefangen.



Gehen Sie bei der bedingten Berechnung der Differenz wie folgt vor:

1. Markieren Sie die Zelle F11 und geben Sie ein Gleichheitszeichen = ein.
2. Wählen Sie nach Eingabe des Buchstabens W die *WENN*-Funktion aus.
3. Die Prüfung, ob die Ist-Kosten eingetragen sind, erfolgt durch Vergleich der entsprechenden Zelle mit "" – ohne Leerzeichen zwischen den Anführungszeichen. Klicken Sie somit die Zelle E11 an – Excel übersetzt diesen Zellenbezug aufgrund der »intelligenten« Tabelle in [@[Ist-Kosten]]. Geben Sie ein Gleichheitszeichen = und anschließend zwei Anführungszeichen "" ein. Schließen Sie das Argument *Prüfung* mit einem Semikolon ab.

Abbildung 4.12 Zelle in Spalte *Ist-Kosten* wird auf leer geprüft

```
=WENN([@[Ist-Kosten]]="");
```

4. Im *Dann_Fall* soll die entsprechende Zelle in der Spalte *Diff. abs.* ebenfalls leer bleiben. Tippen Sie also wiederum zwei Anführungszeichen "" ohne Leerzeichen dazwischen ein und schließen Sie diesen Fall mit Semikolon ab.

Abbildung 4.13 Zelle in Spalte *Diff. abs.* soll leer bleiben, wenn es keine Ist-Kosten in dieser Zeile gibt

```
=WENN([@[Ist-Kosten]]=""; "");
```

5. Im *Sonst_Fall* soll die Berechnung erfolgen. Markieren Sie die Zelle E11, geben Sie ein Minuszeichen - ein und klicken Sie anschließend die Zelle D11 an.

Abbildung 4.14 Zelle in Spalte *Diff. abs.* soll sich bei vorhandenen Ist-Kosten aus der Differenz von Ist- und Sollkosten ergeben

```
=WENN([@[Ist-Kosten]]=""; "", [@[Ist-Kosten]]-[@Soll-Kosten])
```

6. Schließen Sie die Berechnung mit einer schließenden Klammer ab.

Die Formel lautet: =WENN([@[Ist-Kosten]]=""; ""; [@[Ist-Kosten]]-[@Soll-Kosten]) und das Ergebnis ist -15.046 €. Die Ist-Kosten liegen also noch weit unter den Soll-Kosten. Wie in Abbildung 4.11 zu sehen, wird die *Diff. abs.* für alle Projekte mit leeren Ist-Kosten nicht berechnet und die Zelle bleibt somit leer.

Soll- und Ist-Kosten prozentual im Vergleich

In Spalte G *Kosten %* soll – wie in Abbildung 4.11 zu sehen – ein Wert über 100% den prozentualen Über- oder Unterhang der Ist-Kosten im Vergleich zu den Soll-Kosten darstellen. Auch ist eine Berechnung nur dann sinnvoll, wenn bereits Ist-Kosten erfasst wurden.



Gehen Sie wie im letzten Beispiel zur Berechnung der Spalte *Diff. abs.* vor, mit dem Unterschied, dass hier nicht die Differenz der Soll- und Ist-Kosten berechnet wird, sondern der Quotient aus Soll- und Ist-Kosten.



Im *Sonst_Fall* geben Sie also statt eines Minuszeichens ein Divisionszeichen / ein und erhalten letztlich die in Abbildung 4.15 gezeigte Formel.

Abbildung 4.15 Formel zur Ermittlung des prozentualen Vergleichs der Soll- und der Ist-Kosten

```
=WENN([@[Ist-Kosten]]="";"";[@[Ist-Kosten]]/[@[Soll-Kosten]])
```

Die Ergebnisse der Spalte *Kosten %* sind in Abbildung 4.11 zu sehen. Bei fehlenden Ist-Kosten bleiben die Zellen leer.



Kumulierte Differenzen geben Aufschluss über die Gesamtkosten

Interessant sind nicht nur die Kostendifferenzen zwischen Soll- und Istwerten der einzelnen Projekte, sondern auch für die Gesamtheit der Projekte. Da die Projekte nach ihrem Planbeginn sortiert sind, wäre die Information aufschlussreich, wie sich die kumulierten Soll- und Ist-Differenzen projektübergreifend im Lauf der Zeit entwickeln.



Die kumulierten Kostendifferenzen sollen nur bei denjenigen Projekten erscheinen, die auch zur Gesamtbeziehung hinzugezogen werden.



Berechnen Sie diese in der Spalte *kum. Diff.* wie folgt:

1. Markieren Sie die Zelle H11, geben Sie ein Gleichheitszeichen = ein und wählen Sie nach Eingabe des Buchstabens W wiederum die WENN-Funktion aus.
2. Als Argument für *Prüfung* wählen Sie die Zelle F11 aus, geben ein Ungleichheitszeichen <> und anschließend zwei Anführungszeichen " " ein. Schließen Sie das Argument *Prüfung* mit einem Semikolon ab.

Abbildung 4.16 Kumulierte Berechnung soll nur dort erfolgen, wo *Diff. abs.* nicht leer ist

```
=WENN([@[Diff. abs.]<>"");
```

3. Die im nun folgenden *Dann_Wert* zu berechnende Summe reicht immer von der ersten Zeile bis zur aktuellen Zeile der Spalte *Diff. abs.* Ein solcher Bereich kann der Summenfunktion mitgegeben werden, indem der Anfang des Bereichs absolut und das Ende des Bereichs relativ zur aktuellen Zeile gesetzt wird.

Wählen Sie nach Eingabe der Buchstaben SU die Funktion SUMME aus, geben Sie \$F\$11:F11 ein und beenden Sie den *Dann_Wert* mit einer schließenden runden Klammer und einem Semikolon.

Abbildung 4.17 Kumulierte Summenberechnung in Zeilen mit nicht leerer *Diff. abs.*

```
=WENN([@[Diff. abs.]<>"";SUMME($F$11:F11);
```

4. Da im letzten Argument, dem *Sonst_Wert*, keine Berechnung erfolgen soll, geben Sie hier nur zwei Anführungszeichen " " ein.

Abbildung 4.18 Kumulierte Summenberechnung in Zeilen mit leerem *Diff. abs.* aussetzen

```
=WENN([@[Diff. abs.]<>"";SUMME($F$11:F11);"
```

5. Setzen Sie ans Ende der Formel eine schließende runde Klammer.

Sie erhalten die Formel =WENN([@[Diff. abs.]<>"";SUMME(\$F\$11:F11);"") und in Zelle H18 ist das Ergebnis -38.709 €. Wie in Abbildung 4.11 zu sehen ist, wird die kumulierte Differenz nur bedingt gebildet und bezieht auch nur solche Projekte in die Berechnung ein, bei denen Ist-Kosten angegeben sind.

Unschöne Fehlerwerte in Berechnungen ganz einfach vermeiden mit WENNFEHLER

Manchmal fehlen nicht nur Ist-Angaben, sondern auch Soll-Angaben. Wird dann dividiert und ein Divisor ist noch nicht vorhanden, zeigt Excel die unschöne Meldung $\#DIV/0$ an. Solche Fehlermeldungen schließen Sie von vornherein aus, indem Sie die in Excel 2007 hinzugekommene Funktion WENNFEHLER nutzen.

Abbildung 4.19 VORHER: bei fehlender Soll-Dauer zeigt Excel die Fehlermeldung $\#DIV/0$

	A	B	C	I	J	K
10	Projekt	MA	Plan-Beginn	Soll-Dauer	Ist-Dauer	Dauer %
11	Office-Umstellung	12	01.03.2012	150	166	111%
12	Virtualisierung	4	05.03.2012	90		0%
13	TK-Anlage	6	06.03.2012	20	19	95%
14	Project-Umstellung	2	08.03.2012	60	55	92%
15	Windows-Umstellung	5	10.03.2012	85	133	156%
16	Intranet-Update	3	14.03.2012	66		0%
17	SAP-Umstellung	11	15.03.2012	270	283	105%
18	Datensicherheit	7	25.03.2012	160	144	90%
19	Datenbank-Update	2	28.03.2012			$\#DIV/0!$

Abbildung 4.20 NACHHER: die Funktion WENNFEHLER verhindert die unschöne Fehlermeldung $\#DIV/0$

	A	B	C	I	J	K
10	Projekt	MA	Plan-Beginn	Soll-Dauer	Ist-Dauer	Dauer %
11	Office-Umstellung	12	01.03.2012	150	166	111%
12	Virtualisierung	4	05.03.2012	90		0%
13	TK-Anlage	6	06.03.2012	20	19	95%
14	Project-Umstellung	2	08.03.2012	60	55	92%
15	Windows-Umstellung	5	10.03.2012	85	133	156%
16	Intranet-Update	3	14.03.2012	66		0%
17	SAP-Umstellung	11	15.03.2012	270	283	105%
18	Datensicherheit	7	25.03.2012	160	144	90%
19	Datenbank-Update	2	28.03.2012			

In Spalte K *Dauer %* soll ein prozentualer Vergleich zwischen der Ist- und der Soll-Dauer erfolgen. Um die Fehlermeldung $\#DIV/0!$ bei Nichtvorhandensein von Werten auszuschließen, wenden Sie auf die Division Ist-Dauer durch Soll-Dauer die Funktion WENNFEHLER an.



Dies lösen Sie wie folgt:

1. Klicken Sie die Zelle K11 an. Geben Sie ein Gleichheitszeichen = ein und wählen Sie nach Eingabe der Buchstaben WE die Funktion WENNFEHLER aus. Wie in Abbildung 4.21 ersichtlich, benötigt die Funktion zwei Parameter: den Fehler und den Wert im Falle eines Fehlers.



Abbildung 4.21 Die Funktion WENNFEHLER mit zwei Argumenten

`WENNFEHLER(Wert; Wert_falls_Fehler)`

2. Für das Argument *Wert* markieren Sie die Zelle J11, geben das Divisionszeichen / ein und klicken anschließend die Zelle I11 an. Schließen Sie die Eingabe dieses Arguments mit einem Semikolon ab.

Abbildung 4.22 Die WENNFEHLER-Funktion nach Eingabe des *Wert*-Arguments

`=WENNFEHLER([@[Ist-Dauer]]/[@[Soll-Dauer]];`


3. Im Fehlerfall soll keinerlei Berechnung erfolgen und die Zelle leer bleiben. Geben Sie also zwei Anführungszeichen "" ein und schließen Sie die Formel mit einer schließenden runden Klammer und  ab.

Abbildung 4.23 Die komplette WENNFEHLER-Funktion bei der Berechnung der prozentualen Dauer

`=WENNFEHLER([@[Ist-Dauer]]/[@[Soll-Dauer]]; "")`

Die Ergebnisformel lautet `=WENNFEHLER([@[Ist-Dauer]]/[@[Soll-Dauer]]; "")`. Liegt die prozentuale Angabe über 100%, ist die Soll-Dauer überschritten worden. Abbildung 4.20 zeigt das Ergebnis der Berechnung. In Zeilen mit fehlender Soll-Dauer bleibt die Zelle leer.



Die WENNFEHLER-Funktion erspart Ihnen einiges an Schreibarbeit. Vorher hätten Sie deutlich umständlicher die folgende WENN-Abfrage einbauen müssen:

`=WENN(ISTFEHLER([@[Ist-Dauer]]/[@[Soll-Dauer]])); ""; [@[Ist-Dauer]]/[@[Soll-Dauer]]`

Statusinformationen liefern mit WENN



In der Projektliste sollen Bewertungen für die Kosten erfolgen. Die Kosten sollen als »überzogen« gekennzeichnet werden, wenn die Ist-Kosten über den Soll-Kosten liegen, ansonsten als »im Plan«.

Abbildung 4.24 Vorschau auf die fertige Lösung: individuelle Kostenauswertung

	A	B	C	D	E	F	G	H	I
10	Projekt	MA	Plan-Beginn	Soll-Kosten	Ist-Kosten	Diff. abs.	Kosten %	Sum. Diff.	Kosten
11	Office-Umstellung	12	01.03.2012	254.000 €	238.954 €	-15.046 €	94%	-15.046 €	im Plan
12	Visualisierung	4	05.03.2012	66.000 €	0 €	-66.000 €	0%	-66.046 €	im Plan
13	IT-Anlage	6	06.03.2012	114.500 €	120.964 €	6.464 €	105%	-74.882 €	überzogen
14	Projekt-Anstellung	2	08.03.2012	32.400 €	32.400 €	0 €	100%	-74.882 €	im Plan
15	Windows-Umstellung	5	10.03.2012	93.800 €	108.400 €	14.600 €	116%	-60.282 €	überzogen
16	Intranet-Update	3	11.03.2012	67.400 €					im Plan
17	SAP-Umstellung	11	15.03.2012	342.100 €	362.743 €	20.643 €	106%	-39.639 €	überzogen
18	Datensicherheit	7	25.03.2012	159.300 €	160.250 €	950 €	101%	-38.709 €	überzogen
19	Datenbank-Update	2	28.03.2012	32.490 €					im Plan

Um mithilfe von WENN-Funktionen die Bewertung der Kosten zu testen, öffnen Sie die Datei *Kap_04_Logik_UEB.xlsx*. Wechseln Sie dort zum Arbeitsblatt *WENN verschachtelt*.



Führen Sie die Bewertung der Kosten folgendermaßen aus:

1. Markieren Sie die Zelle L11, geben Sie ein Gleichheitszeichen = sowie den Buchstaben W ein und wählen Sie dann die WENN-Funktion aus.
2. Zur Angabe der *Prüfung* klicken Sie die Zelle E11 an, geben ein Größer- Zeichen > ein und markieren anschließend die Zelle D11. Schließen Sie dieses Argument mit einem Semikolon ab.



Abbildung 4.25 Bedingung eingeben für die Bewertung der Ist- und Soll-Kosten mit der WENN-Funktion

```
=WENN([@[Ist-Kosten]]>@[Soll-Kosten]);
```

3. Die Bewertung im *Dann_Wert* soll mit dem festen Text *überzogen* erfolgen. Geben Sie diesen Text in Anführungszeichen " " ein und schließen Sie das Argument mit Semikolon ab.

Abbildung 4.26 Falls die Ist-Kosten über den Soll-Kosten liegen, wird der Text *überzogen* angezeigt

```
=WENN([@[Ist-Kosten]]>@[Soll-Kosten]; "überzogen";
```

4. Im *Sonst_Wert* soll der Text *im Plan* ausgegeben werden. Geben Sie diesen Text wieder von Anführungszeichen " " ummantelt ein und schließen Sie die Funktion mit einer schließenden runden Klammer ab.

Abbildung 4.27 Vollständige Formel zur Bewertung des Vergleichs von Ist- und Soll-Kosten

```
=WENN([@[Ist-Kosten]]>@[Soll-Kosten]; "überzogen"; "im Plan")
```

Im Ergebnis erscheint die Formel wie in Abbildung 4.27 dargestellt und in Zelle L11 das Ergebnis im Plan. Alle Ergebnisse der Spalte *Kosten* sind in Abbildung 4.24 zu sehen.

Mehr als zwei mögliche Ergebnisse durch Verschachteln von WENN

Es ist möglich, weitere WENN-Funktionen in den *Dann_Wert* und/oder in den *Sonst_Wert* einzufügen (Verschachtelungen). Es können bis zu 64 Argumente auf diese Weise verschachtelt werden, aber es müssen viele Klammern gesetzt werden.

Bei den Terminen soll eine Einstufung in drei Fälle erfolgen:

- Liegt die oben berechnete prozentuale Dauer unter 100%, ist zu unterscheiden, ob das Projekt noch läuft und damit als *im Plan* gekennzeichnet wird.



- Falls nicht, soll *pünktlich* erscheinen.
- Ist aber der Prozentwert für die Dauer größer oder gleich 100%, so gilt das Projekt als *überzogen*.

Abbildung 4.28 In der Vorschau sind Zustände für Termine zu sehen: *überzogen*, *pünktlich* oder *im Plan*

	A	B	C	D	E	F	G	H	I	J	K	L	M
10	Projekt	MA	Plan-Beginn	Soll-Dau	Ist-Dauer	Dauer %							Termine
11	Office-Umstellung	12	01.03.2012	150	166	111%							überzogen
12	Virtualisierung	4	05.03.2012	90		0%							im Plan
13	TK-Anlage	6	06.03.2012	20	19	95%							pünktlich
14	Project-Umstellung	2	08.03.2012	60	55	92%							pünktlich
15	Windows-Umstellung	5	10.03.2012	85	133	156%							überzogen
16	Intranet-Update	3	14.03.2012	66		0%							im Plan
17	SAP-Umstellung	11	15.03.2012	270	283	105%							überzogen
18	Datensicherheit	7	25.03.2012	160	144	90%							im Plan
19	Datenbank-Update	2	28.03.2012										überzogen




Blieben Sie in der Datei *Kap_04_Logik_UEB.xlsx* im Arbeitsblatt *WENN verschachtelt*, um die Bewertung der Termine vorzunehmen.



Gehen Sie bei der Lösung der Aufgabe wie in *Abbildung 4.29* skizziert vor.

Abbildung 4.29 Die Struktur der verschachtelten WENN-Funktionen




Um mit der Klammersetzung nicht durcheinanderzugeraten, können Sie auch den Funktionsassistenten bei der Erstellung der Formel verwenden. Dies geschieht in folgenden Schritten:

1. Markieren Sie die Zelle M11, geben Sie das Gleichheitszeichen = sowie WE ein und wählen Sie die Funktion *WENN* aus.
2. Als Bedingung für die erste WENN-Funktion prüfen Sie, ob die prozentuale Dauer unter 100% liegt. Excel versteht an dieser Stelle die Eingabe des Prozentzeichens. Klicken Sie somit die Zelle K11 an, geben Sie das Kleiner-Zeichen < ein und anschließend 100%. Schließen Sie die Prüfung mit einem Semikolon ab.

- Im *Dann_Wert* – prozentuale Dauer unter 100% – soll weiter unterschieden werden, ob der Plan-Beginn plus die Soll-Dauer des Projekts ein Datum vor dem heutigen Tag ergibt. Somit geben Sie an dieser Stelle ein neues WENN() ein.
- Klicken Sie nun auf das Symbol *Funktion einfügen (fx)* links der Bearbeitungsleiste an und geben Sie im folgenden Dialogfeld – wie in Abbildung 4.30 gezeigt – die *Prüfung*, den *Dann_Wert* und den *Sonst_Wert* für die zweite WENN-Funktion ein.

Abbildung 4.30 Eingabe der Argumente für die WENN-Funktion im Funktionsassistenten



- Wie oben beschrieben, markieren Sie nun für die Prüfung die Zelle C11, geben das Pluszeichen + ein und klicken dann die Zelle I11 an. Das sich daraus ergebende Datum vergleichen Sie mit dem heutigen Datum, das Sie in Excel mit der Funktion HEUTE() erhalten. Geben Sie also das Kleiner-Zeichen < ein und tippen HEUTE(). Die Prüfung lautet: [@[Plan-Beginn]]+[@[Soll-Dauer]]<HEUTE()
- Im *Dann_Wert* geben Sie pünktlich ein. Hier können Sie sich die Anführungszeichen sparen; der Assistent setzt sie automatisch.
- Im *Sonst_Wert* tippen Sie im Plan ein und bestätigen das Dialogfeld mit OK.
- Eine eventuell angezeigte Fehlermeldung wegen einer fehlenden Klammer bestätigen Sie mit OK. Klicken Sie in der Bearbeitungsleiste ans Ende der Formel, geben Sie ein Semikolon ein und anschließend in Anführungszeichen " " den Text überzogen. Schließen Sie die Formel mit einer schließenden Klammer ab.

Die gesamte Formel sieht dann wie in Abbildung 4.31 gezeigt aus:

Abbildung 4.31 Zwei geschachtelte WENN-Funktionen

```
=WENN([@[Dauer %]]<100%;WENN([@[Plan-Beginn]]+[@[Soll-Dauer]]<HEUTE();
"pünktlich";"im Plan");"überzogen")
```

Die Ergebnisse der Spalte *Termine* sind in Abbildung 4.28 zu sehen.

Noch mehr Bedingungen definieren mit UND, ODER und NICHT

Bei den bisherigen Prüfungen mit der WENN-Funktion ging es stets um *eine* Bedingung. Sollen mehrere Bedingungen kombiniert werden, um umfangreichere Prüfungen für die Auswertung vornehmen zu können, kommen die Funktionen UND, ODER und NICHT ins Spiel.

Die UND-Funktion überprüft bis zu 255 Argumente auf ihren Wahrheitsgehalt. Der Wahrheitswert WAHR wird nur dann geliefert, wenn alle Bedingungen zutreffen. Treffen eine oder mehrere Bedingungen nicht zu, wird der Wahrheitswert FALSCH zurückgegeben.

Abbildung 4.32 Die Funktion UND gibt nur dann WAHR zurück, wenn alle Bedingungen zutreffen



In der WENN-Funktion tritt also der *Dann_Wert* nur in Kraft, wenn alle Prüfungsergebnisse WAHR sind. Ergibt die Prüfung in einem oder mehreren Fällen FALSCH, kommt der *Sonst_Wert* zum Tragen.


Im Unterschied dazu überprüft die Funktion ODER zwar auch bis zu 255 Werte auf ihren Wahrheitsgehalt, der Wahrheitswert FALSCH wird allerdings nur dann geliefert, wenn alle Bedingungen nicht zutreffen. Anders gesagt: Trifft eine Bedingung zu, liefert diese Funktion bereits WAHR zurück.

Abbildung 4.33 Die Funktion ODER gibt nur dann FALSCH zurück, wenn alle Bedingungen nicht zutreffen



Die Funktion NICHT kehrt das Ergebnis einer Prüfung um. So ergibt =NICHT(WAHR) den Wert FALSCH. NICHT wird in der Prüfung der WENN-Funktion häufig verwendet, um Zustände auszuschließen.

Abbildung 4.34 Die Funktion NICHT kehrt den Wahrheitswert des Arguments um



Stellen Sie sich die Frage, welchen Wert die Funktion NICHT liefert, wenn ihr als Argument ein Zahlenwert übergeben wird. Sie liefert den Wert WAHR, wenn die übergebene Zahl eine 0 ist. Probieren Sie es einmal selbst aus, indem Sie in eine leere Zelle =NICHT(0) eingeben.

Positiv laufende Projekte herausstellen

In der Projektübersicht sollen für einen schnellen Gesamtüberblick die positiv laufenden Projekte mit dem Symbol einer Sonne gekennzeichnet werden. Dies ist der Fall, wenn sowohl die Kosten als auch die Termine als nicht *überzogen* gekennzeichnet wurden.



Abbildung 4.35 Positiv laufende Projekte durch eine Sonne gekennzeichnet

	A	B	C	L	M	N	O
10	Projekt	MA	Plan-Beginn	Kosten	Termine	Gesamt	Status
11	Office-Umstellung	12	01.03.2012	Im Plan	Überzogen		Achtung
12	Virtualisierung	4	05.03.2012	Im Plan	Im Plan	☀	
13	TK-Anlage	6	06.03.2012	Überzogen	pörrtlich		
14	Project-Umstellung	2	08.03.2012	Im Plan	pörrtlich	☀	
15	Windows-Umstellung	5	10.03.2012	Überzogen	Überzogen		
16	Intranet-Update	3	14.03.2012	Im Plan	Im Plan	☀	
17	SAP-Umstellung	11	15.03.2012	Überzogen	Überzogen		Achtung
18	Datensicherheit	7	25.03.2012	Überzogen	Im Plan		Achtung
19	Datenbank-Update	2	28.03.2012	Im Plan	Überzogen		


Sie können dieses Beispiel in der Datei *Kap_04_Logik_UEB.xlsx* im Arbeitsblatt *WENN-UND-ODER-NICHT* nachvollziehen.



Orientieren Sie sich beim Aufstellen der Formel an dem in Abbildung 4.36 dargestellten Ablauf.



Abbildung 4.36 Das Symbol »Sonne« anzeigen, wenn Kosten und Termine nicht überzogen sind



Gehen Sie folgendermaßen vor:

1. Markieren Sie die Zelle N11, geben Sie ein Gleichheitszeichen = sowie WE ein und wählen Sie dann die *WENN*-Funktion aus.
2. Geben Sie ein Semikolon ; ein und anschließend *ZEICHEN(15)*, um diesmal zunächst den *Dann_Wert* zu bestimmen. *ZEICHEN(15)* liefert als Symbol die Sonne.
3. Geben Sie ein weiteres Semikolon an, um auch den *Sonst_Wert* zu bestimmen. Dieser soll einfach leer sein. Geben Sie deshalb zwei Anführungszeichen "" hintereinander und dann die schließende runde Klammer ein.


Abbildung 4.37 Den Rahmen für eine WENN-Funktion mit umfangreicher Prüfung schaffen

```
=WENN( ; ZEICHEN(15); "" )
WENN(Prüfung; [Dann_Wert]; [Sonst_Wert])
```

4. Setzen Sie nun den Cursor in der Bearbeitungsleiste nach WENN(, um die Bedingung für die Prüfung einzugeben. Geben Sie die folgende Formel ein:

```
UND(NICHT([@Kosten]="überzogen"); NICHT([@Termine]="überzogen"))
```

Den Text [*@Kosten*] erstellt Excel automatisch, wenn Sie an dieser Stelle die Zelle L11 anklicken; den Text [*@Termine*] durch Anklicken der Zelle M11.

5. Schließen Sie die Formel mit  ab.

Die komplette Formel ist in Abbildung 4.38 zu sehen. Sie liefert für die Projekte in den Zeilen 12, 14 und 16 als positive Rückmeldung zum Gesamtverlauf des Projekts das Symbol einer Sonne.

Abbildung 4.38 WENN-Funktion mit umfangreicher Prüfung

```
=WENN(UND(NICHT([@Kosten]="überzogen"); NICHT([@Termine]="überzogen"));
ZEICHEN(15); "")
```



Der ZEICHEN-Funktion kann eine Zahl von 1 bis 255 mitgegeben werden, die das von Ihnen gewünschte Zeichen angibt. Das jeweilige Zeichen ist Bestandteil des Zeichensatzes, der auf Ihrem Computer verwendet wird. So liefert die ZEICHEN-Funktion mit dem Argument 15 eine Sonne. Eine Zusammenstellung aller möglichen darstellbaren Zeichen ist in der Datei *ZEICHEN.XLSX* hinterlegt.

Diverse Kriterien zu einer Entscheidung zurate ziehen

Um verschiedene Bedingungen zusammenzuführen, werden nicht nur die UND- und die NICHT-Funktion verwendet, sondern auch die ODER-Funktion. So können Sie diverse Informationen in der Prüfung kombinieren, die dann letztendlich wieder einen der Wahrheitswerte WAHR oder FALSCH liefern.

Beachten Sie bei der UND- sowie bei der ODER-Funktion, dass in unserem Sprachgebrauch diese beiden Begriffe anders benutzt werden. Beispielsweise sollen in einer Mailingaktion alle Lieferanten aus Darmstadt **und** Frankfurt angeschrieben werden. Da die Lieferanten aber nicht gleichzeitig aus Darmstadt und Frankfurt kommen, sind es Lieferanten aus Darmstadt **oder** Lieferanten aus Frankfurt.



Im allgemeinen Sprachgebrauch wird »UND« gesprochen und geschrieben, gemeint ist aber »ODER«.

Bei der ODER-Funktion muss – wie oben bereits erläutert – mindestens eine Bedingung zutreffen, damit das Ergebnis WAHR geliefert wird. In Kombination mit der WENN-Funktion bedeutet dies, dass der *Dann_Wert* verwendet wird, wenn auch nur eine Bedingung wahr ist. Trifft keine Bedingung zu, tritt der *Sonst_Wert* in Kraft.

- Tippen Sie ein weiteres Semikolon ein und danach zwei Anführungszeichen "" für den leeren *Sonst_Wert*. Beenden Sie die WENN-Funktion mit einer schließenden Klammer.

Abbildung 4.41 Den Rahmen für eine WENN-Funktion mit umfangreicher Prüfung schaffen



- Setzen Sie den Cursor nach WENN(in die Bearbeitungsleiste, denn nun folgt die Eingabe der Prüfung, die sich aus UND- sowie ODER-Funktionen zusammensetzt. Erfassen Sie dort die folgende Formel:

```
ODER(UND([@[Soll-Kosten]]>150000;[@[Kosten %]]>100%);
UND([@[Soll-Dauer]]>100;[@[Dauer %]]>100%))
```



Alle mit @ gekennzeichneten Zellenangaben können Sie ganz einfach erreichen, indem Sie in Zeile 11 die Zelle in der entsprechenden Spalte anklicken. So sparen Sie sich unnötige Tipparbeit und schließen Tippfehler aus.

- Schließen Sie die Bearbeitung der Formel mit ab.

Abbildung 4.42 Eine umfangreiche UND/ODER-Formel in der Prüfung einer WENN-Funktion



Die abschließende Formel sieht nun wie in Abbildung 4.42 gezeigt aus und gibt Ihnen ein deutliches Signal bei den kritischen Projekten in den Zeilen 11, 17 und 18.

In Tabelle 4.2 finden Sie eine Zusammenstellung von häufig benutzten Funktionen zur Prüfung der Bedingung.

Tabelle 4.2 Häufig benutzte Funktionen zur Prüfung der Bedingung

Funktion	Bedeutung	Beispiel	Ergebnis
UND(Bed1;Bed2;...)	WAHR, wenn alle Bedingungen wahr sind, sonst FALSCH	=UND(WAHR;FALSCH) =UND(WAHR;WAHR) =UND(1=1;2=3) =UND(1=1;2<3)	FALSCH WAHR FALSCH WAHR
ODER(Bed1;Bed2;...)	WAHR, wenn mindestens eine Bedingung wahr ist, sonst FALSCH	=ODER(WAHR;FALSCH) =ODER(FALSCH;FALSCH) =ODER(1=1;2=3) =ODER(1<1;2>3)	WAHR FALSCH WAHR FALSCH
NICHT(Wahrheitswert)	kehrt den Wahrheitswert um	=NICHT(WAHR) =NICHT(1=2)	FALSCH WAHR

Ein weiteres Beispiel für die Kombination von WENN und UND

Falls Sie sich für weitere Anwendungsgebiete der Funktion WENN interessieren, finden Sie in der Beispieldatei *Kap_04_Logik_LOE.xlsx* im Arbeitsblatt *Zusatz* ein Beispiel, das Sie in abgewandelter Form vielleicht für die Arbeit in Vereinen nutzen können. Abbildung 4.43 zeigt eine Vorschau auf die Lösung, in der wiederum die UND-Funktion innerhalb der WENN-Abfrage verwendet wird.

Abbildung 4.43 Möglicherweise können Sie diese Lösung für die eigene Vereinsarbeit anpassen

	A	B	C	D	E
11	Abstimmung				
12	Beschlussvorlage	Mitglieder	davon anwesend	Ja-Stimmen	Beschluss
13	A	210	100	99	abgelehnt
14	B	210	125	88	angenommen
15	C	210	160	102	abgelehnt
16	D	210	99	50	abgelehnt
17	E	210	170	155	angenommen

Die Ergebnisse einer WENN-Funktion können Sie übrigens auch in unterschiedlichen Farben oder mit verschiedenen Symbolen darstellen. Mehr dazu erfahren Sie in Kapitel 8, in dem es um die bedingte Formatierung geht.

